
Objects as Methods Simulation
Context≫send: selector to: rcvr with: arguments lookupIn: lookupClass

"Simulate the action of sending a message with selector and arguments to rcvr. The argument, lookupClass, is the class in which to lookup the message. 
This is the receiver's class for normal messages, but for super messages it will be some specific class related to the source method."

| meth primIndex val ctxt |
(meth := lookupClass lookupSelector: selector) ifNil:
[selector == #doesNotUnderstand: ifTrue:

[self error: 'Recursive message not understood!' translated].
^self send: #doesNotUnderstand:

to: rcvr
with: {(Message selector: selector arguments: arguments) lookupClass: lookupClass}
lookupIn: lookupClass].

meth isCompiledMethod ifFalse:
["Object as Methods (OaM) protocol: 'The contract is that, when the VM encounters an ordinary object (rather than a compiled method) in the 

method dictionary during lookup, it sends it the special selector #run:with:in: providing the original selector, arguments, and receiver.'. DOI: 
10.1145/2991041.2991062."

^self send: #run:with:in:
to: meth
with: {selector. arguments. rcvr}].

meth numArgs = arguments size ifFalse:
[^ self error: ('Wrong number of arguments in simulated message {1}' translated format: {selector})].

(primIndex := meth primitive) > 0 ifTrue:
[val := self doPrimitive: primIndex method: meth receiver: rcvr args: arguments.
(self isPrimFailToken: val) ifFalse:

[^val]].

(selector == #doesNotUnderstand: and: [lookupClass == ProtoObject]) ifTrue:
[^self error: ('Simulated message {1} not understood' translated format: {arguments first selector})].

ctxt := Context sender: self receiver: rcvr method: meth arguments: arguments.
(primIndex isInteger and: [primIndex > 0]) ifTrue:
[ctxt failPrimitiveWith: val].

^ctxt



Process-Faithful Debugging

› Processor activeProcess

evaluate: [

self error.

self inform: #foo]

onBehalfOf: [] newProcess

› Processor activeProcess

environmentAt: #foo put: 42.

Processor activeProcess

environmentAt: #foo.  → 42

› Process≫evaluate: aBlock onBehalfOf: aProcess

"Evaluate aBlock setting effectiveProcess to 

aProcess. Used in the execution simulation 

machinery to ensure that Processor 

activeProcess evaluates correctly when 

debugging.“

| oldEffectiveProcess |

oldEffectiveProcess := effectiveProcess.

effectiveProcess := aProcess.

^ aBlock ensure: [

effectiveProcess := oldEffectiveProcess]

› ProcessorScheduler≫activeProcess

"Answer the currently running Process.“

^ activeProcess effectiveProcess



Process-Faithful Debugging – Patch

StandardToolSet≫handleError: anError
"Double dispatch. Let the active process take 

care of that error, which usually calls back here to 
#debugProcess:..."

^ Processor basicAactiveProcess
debug: anError signalerContext
title: anError description



Debugger Chains



Context≫#runUntilErrorOrReturnFrom:

Context≫runUntilErrorOrReturnFrom: aSender

"ASSUMES aSender is a sender of self. Execute self's stack until aSender returns or an unhandled exception is raised. Return a pair containing the new top context and a possibly nil exception. The exception is not nil if it 

was raised before aSender returned and it was not handled. The exception is returned rather than openning the debugger, giving the caller the choice of how to handle it."

"Self is run by jumping directly to it (the active process abandons thisContext and executes self). However, before jumping to self we insert an ensure block under aSender that jumps back to thisContext when evaluated. 

We also insert an exception handler under aSender that jumps back to thisContext when an unhandled exception is raised. In either case, the inserted ensure and exception handler are removed once control jumps back to 

thisContext."

| error ctxt here topContext |

here := thisContext.

"Insert ensure and exception handler contexts under aSender"

error := nil.

ctxt := aSender insertSender: (Context

contextOn: UnhandledError do: [:ex |

error

ifNil: [

error := ex exception.

topContext := thisContext.

ex resumeUnchecked: here jump]

ifNotNil: [ex pass] ]).

ctxt := ctxt insertSender: (Context

contextEnsure: [error ifNil: [

topContext := thisContext.

here jump] ]).

self jump. "Control jumps to self"

"Control resumes here once above ensure block or exception handler is executed"

^ error

ifNil: [

"No error was raised, remove ensure context by stepping until popped"

[ctxt isDead] whileFalse: [topContext := topContext stepToCallee].

{topContext. nil}]

ifNotNil: [

"Error was raised, remove inserted above contexts then return signaler context"

aSender terminateTo: ctxt sender. "remove above ensure and handler contexts"

{topContext. error}]

Generator on: [:stream |

stream nextPut: #foo]



Further reading :-)

› Thread on “Debugger chains”: [squeak-dev] Fixing the 
infinite debugger chains?

› Simulation of Objects as Methods: Kernel-ct.1357 (Trunk)

› Thread on sender swaps and #runUntilErrorOrReturnFrom:: 
[squeak-dev] BUG/REGRESSION while debugging 
Generator >> #nextPut:

› In your image …:
– Debugger>>step{Over, Through}

– Process>>evaluate:onBehalfOf:

– {Process, Context}>>step:

– Context>>runUntilErrorOrReturnFrom:

http://forum.world.st/I-broke-the-debugger-tp5110752p5110814.html
http://forum.world.st/The-Trunk-Kernel-ct-1357-mcz-td5124373.html
http://forum.world.st/BUG-REGRESSION-while-debugging-Generator-nextPut-tp5108125.html

